
Лекция 3. Указатели. Списки 

 

Новый тип данных, с которым связаны все динамические типы и даже 

статические – Указатель. Значением переменной типа Указатель является Адрес. 

Все ячейки памяти пронумерованы, номер первой из нескольких ячеек памяти, 

занятой данными и является адресом, по которому расположены некоторые данные 

или код в оперативной памяти (скорее – виртуальной оперативной памяти, но об 

этом узнаете при изучении операционных систем, что не является предметом 

изучения данного курса). Хотя пронумерована в компьютере не только оперативная 

память, значениями указателя будут именно адреса оперативной памяти, той ее 

части, что выделена для вашей задачи при ее запуске. 

 

 

 NiL – нулевое значение типа указателя. Говорит о том, что указатель ни на что 

не указывает в данный момент. Для выделения памяти из Кучи (нераспределенной 

памяти, оставшейся от памяти, выделенной данной программе, и потраченной на 

стек вызовов, размещения данных и кода этой программы) можно воспользоваться  

несколькими процедурами: 

 

 Если память была выделена одной из этих процедур, то для освобождения 

следует воспользоваться процедурой FreeMem, а затем указателю присвоить 

значение Nil. 

 Если просто присвоить Nil, то память останется занятой. И потеряв ее адрес, 

уже нельзя будет ее освободить (до завершения программы работы, когда вся 

выделенная программе память станет свободной), чтобы выделить для хранения 

других данных. Это называется «утечкой памяти». При многократной утечке памяти 

может оказаться, что программе невозможно дальше продолжать работать, так как 

не достаточно свободной памяти. И она будет аварийно закрыта, или потребуется ее 

перезапуск. 

 Символ ^ иногда на письме обозначается . И используется не только для 

объявления типизированного указателя, но и (как видно из примера выше) для его 

разыменования, то есть обращения к данным, лежащим по указанному в указателе 

адресу. Для правильной интерпретации этих данных и для того, чтобы знать 

количество байт, на первый из которых указывает указатель, надо привести тип 

нетипизированного указателя к типизированному, например – PInteger (тип 

указателя на 4-хбайтовое целое со знаком, уже объявлен в Delphi, но можно 

определить и самим. Типы указателей обычно начинают с буквы P – от слова 

Pointer). 

 Для типизированного указателя можно использовать еще одну пару процедур – 

New и Dispose, которые не требуют указания размера выделяемой памяти, а 

вычисляют ее по типу данных, на который указывает типизированный указатель. 

 

 Кроме того, адрес можно «позаимствовать» у другой переменной, 

воспользовавшись операцией @ взятия адреса или функцией Addr: 



 Указатели. Списки  Стр.4 из 39 

 

 Указатель может указывать не только на простой тип, например: 

 

 Для освобождения памяти, выделенной с помощью New, воспользуемся 

Dispose: 

 Совместимость типов при присваивании: 

Пример программы с указателями, демонстрирующий работу с разными их 

типами: 
program PrjUkazat; 

 

{$APPTYPE CONSOLE} 

 

uses 

  SysUtils; 

 

Type 

  Tmatr = record 

    n,m: byte; 

    elem: array [1..10, 1..10] of integer; 

  end; 

 

Var 

  k: ^Integer; //типизованный указатель 

  p: Pointer;  // нетипизированный указатель 

  d: ^Integer; // еще один типизированный указатель 

 

  a: Tmatr; // запись: матрица и ее размеры 

  b: ^Tmatr;// указатель на запись, со статическим массивом в ней 

 

  i,j: byte; 

begin 

  try 

    {------------k,d,p-------------} 

    new(k); // выделяем память 

    k^:=77; // присваиваем значение по адресу, хранящемуся в k 

    writeln(k^); // вывод. операция ^ - разыменование указателя 

 

    //d:=k; напрямую нельзя – разные неименованные пользовательские 

типы 

    p:=k;  d:=p; // через нетипизированный указатель 

 

    writeln(d^); // адрес указателя скопирован из k  

 

    d^:=55; // изменяет значение по адресу d (и k) 

    writeln(k^); 

    writeln(d^); 

 

 // вывод адресов переменных-указателей  

  // с взятием адреса двумя способами - @ и Addr 

    writeln(cardinal(@k), '=', Cardinal(Addr(k))); 

    writeln(cardinal(@p), '=', Cardinal(Addr(p))); 

    writeln(cardinal(@d), '=', Cardinal(Addr(d))); 

 // вывод адресов переменных целого типа 

    writeln(cardinal(@i), '=', Cardinal(Addr(i))); 

    writeln(cardinal(@j), '=', Cardinal(Addr(j))); 

 // вывод значений(адресов) указателей 

    writeln(cardinal(k), '=', Cardinal(d)); 

 

    Dispose(k); // не присваивается nil 

    writeln(k^);  

    writeln(d^); 

    k:=nil; 

    writeln; 

 

    {--------------matr-------------} 

    randomize; 

     

    a.n:=2;  // заполняем массив а и его размеры 

    a.m:=3; 

    for i:=1 to a.n do 

      for j:=1 to a.m do a.elem[i,j]:= random (MaxInt); 

 

    writeln('A (',a.n,'x',a.m,')'); 

    for i:=1 to a.n do 

    begin 

      for j:=1 to a.m do write(a.elem[i,j]:12); 

      writeln; 

    end; 

 

    new(b); // указатель на запись 

    b^.n:=a.n;  // значения берем из а 

    b^.m:=a.m; 

    for i:=1 to b^.n do 

      for j:=1 to b^.m do b^.elem[i,j]:=a.elem[i,j]; 

 

    writeln('B (',b^.n,'x',b^.m,')'); 

    for i:=1 to b^.n do 

    begin 

      for j:=1 to b^.m do write(b^.elem[i,j]:12); 

      writeln; 

    end; 

 

    dispose(b); 

 

  except 

    on E:Exception do 

      Writeln(E.Classname, ': ', E.Message); 

  end; 

 

  readln 

end. 

С помощью указателей можно парой способов создать свой тип динамического 

массива: 

 

 Максимальная длина описываемого типа статического массива – 2Гбайта. 

Указатель на него будет динамическим массивом, и такой способ позволяет 

использовать индексы, причем – с 1, если с 1 описаны индексы в типе. 

 Хотя указатель – типизированный, воспользуемся GetMem, чтобы выделить 

память не на весь массив, а только на N элементов: 

 

 Введем для примера значения в этот массив. Когда массив станет не нужен, 

освободим память. 

 Можно описывать более короткий массив, так как по умолчанию в Delphi 

отключен контроль за диапазонами, и надо лишь обозначить тип элементов массива 

и значение первого индекса): 

Второй способ поможет понять как организованы массивы и строки в Си, 

который нам  предстоит изучать. Там массив – это указатель на первый его элемент: 

 

 Выделение памяти и ее освобождение выглядит также, а вот обращение к 

элементам получается длинным, поскольку требуется преобразование типов. Зато 

становится понятным, как устроен доступ по индексу к элементу массива и почему 

он отсчитывается с нуля. Индекс – это количество элементов, которые надо 

отступить от начала массива, чтобы найти нужный элемент. Описание 

преобразования: 

1) В DynAr хранится адрес первого элемента массива, преобразуем его в число 

типа Cardinal (целое, 4 байта, беззнаковое для 32-хбитного адреса); 

2) Полученное целое число увеличиваем на длину i элементов массива; 

3) Вычисленную сумму - адрес i-го элемента, преобразуем к типу указателя на 

элемент нужного типа; 

4) Разыменование (^) указателя типизированного типа, для доступа к памяти, в 

которой (будет) хранится значение i-го элемента. 

5) В эту память и считываем значение типа real  с помощью процедуры ReadLn. 

 

 Когда массив будет не нужен,  память надо будет освободить: 

 При последовательном переборе элементов, можно высчитывать адрес 

элемента не с адреса первого элемента, а с адреса предыдущего. Для этого удобно 

воспользоваться процедурой Inc, которая помимо порядковых типов, работает и с 

адресами. 

 

 Для создания своего динамического массива важно вспомнить, что 

двухмерный массив – это одномерный массив, каждый элемент которого – 

одномерный массив. 

 Или можем воспользоваться знанием о последовательном хранении в памяти 

элементов одномерного массива, каждый элемент которого – тоже массив с 

последовательным хранением элементов в смежных ячейках памяти. И получим 

массив из последовательно хранящихся строк: 

 В этом примере две строки из трѐх элементов. Для нахождения i-й строки (c 0) 

надо пропустить все три элемента каждой из предыдущих i строк. А затем еще и j 

(тоже с 0) элементов в этой строке, чтобы найти [i,j]-элемент массива. 

 При последовательном переборе всех элементов можно искать каждый 

следующий элемент после предыдущего:  

Рекурсивный тип данный –  список 

Описание типа «Односвязный линейный список» требует использования двух 

взаимно рекурсивных типов (кратко: ссылаются при описании друг на друга, 

подробнее дальше в лекциях про рекурсию) – PElem  и TElem. Это единственная 

ситуация, когда при описании типа (PElem) можно сослаться на тип (TElem), не 

описанный выше по коду: 

 

Без использования указателя его описать невозможно, так как тип должен 

однозначно определять размер памяти, которая будет необходима для хранения 

данных этого типа: 

Описать список можно одной переменной (указатель на первый элемент 

списка) или совокупностью трѐх (на первый, последний, текущий): 

В среде разработке приложений –  Borland Developer Studio 2006 – можно при 

отладке просмотреть содержимое элементов списка, «переходя» по ссылкам: 

 

В шестнадцатеричном виде записаны 32-битные адреса, в 10 и 16 системах 

счисления приводятся значения элементов (в данном случае – целых чисел). 

Сравните со списком справа: 

 

 



 Указатели. Списки  Стр.17 из 39 

Можно сделать ДВЕ связи в списке. Для линейного списка это будут связи с 

предыдущим и следующим элементами: 

При обращение ко второму элементу возникнет ошибка, если нет даже 

первого. И ошибка возникнет при попытке присвоить значение в элемент, для 

хранения которого не выделена память (присваивание по адресу Nil). 

Двусвязный список тоже можно посмотреть в среде разработки во время отладки: 

 

Рассмотрим каждую из операций для работы со списком: 

 

В примере к Лабораторной №3 (в конце этого файла) это будет сделано в 

секции Initialization модуля UnVar. 

Про добавлении в начало еще одного элемента. Надо для него сначала 

выделить память, 

 

а потом связать со списком 

Аналогично, при добавлении в конец сначала выделяем память 

 

Потом связываем со списком 

 

При добавлении в конец можно обойтись без дополнительной переменной, 

сразу создавая новый элемент после последнего: 

Добавление в середину (после текущего, с адресом, хранящимся в ListC) : 

Удаление (с освобождением памяти!) элементов происходит по-элементно. 

Подготовка к циклу: 

 

Подготовка к удалению («старый» первый :уже не в списке) 

 

Удаление (освобождение памяти «старого» первого элемента): 

И подготовка к следующей итерации: 

 

Список стал на 1 элемент короче. 

Выполняем тело цикла, пока список не станет пустым: 

 

Для работы с элементами списка используется последовательный перебор 

элементов. Например, так выглядит цикл перебора элементов списка с помощью 

указателя ListC на «текущий» элемент: 

Найдем с помощью него сумму элементов списка. Поиск сумы начинается с 0. 

Затем, последовательно перебирая элементы списка, увеличиваем сумму, накапливая 

ее: 

Достоинства массивов и списков прямо противоположны:  

Статичный (фактически, при удалении/добавление создается копия всех 

элементов с нужным изменением, часто с перевыделением памяти) массив с 

быстрым произвольным доступом к элементам по индексам 

Динамичный (добавление/удаление быстрое и затрагивает только соседние 

элементы) список с медленным последовательным доступом 

Для каждой задачи удобно выбирать что-то одно, или создавать их «смесь» как 

это сделано при хранении файлов. И как это делается при хранении длинных 

списков, создавая Хэш-функции, позволяющие перемещаться по списку не только 

последовательно, но и произвольно переходить к нужной части упорядоченного 

списка. При выборе эффективной Хэш-функции надо стремиться к тому, чтобы для 

каждого значений функции длины списков были примерно равными. Например, если 

в списке хранятся фамилии, то Хэш-функция по первой букве этих фамилий даст 

очень не равномерный разброс: на какие-то буквы вообще не будет фамилий, а на 

другие, наоборот, их будет несоизмеримо больше: 

Информационно-логические структуры: 

Список односвязный и двусвязный 

 
 

 

 
 

Данные: 

Адрес начала списка, (обязательное данное, остальные для удобства) 

Адрес текущего элемента,  

Адрес конца списка. 

Допустимые операции: 

Создать список из 1 элемента; 

Добавить элемент в начало/конец списка или после текущего (в упорядоченном списке – в 

соответствующее место по указанному порядку); 

Удалить первый, текущий, последний элемент; 

Перейти к следующему/предыдущему/первому/последнему от текущего элемента элементу; 

Узнать значение информационной части; 

Проверка на отсутствие элементов (пустой список); 

Очистить список (удалить все элементы) 

Для работы с элементами списка используется перемещение по списку. 

Очередь  

(Информационно – логическая структура) 

(например, на основе связного линейного списка + порядок доступа FIFO) 

 (FIFO - First In First Out) 

 
Данные: 

Адрес головы очереди,  

Адрес хвоста очереди. 

Допустимые операции: 

Создать очередь из 1 элемента; 

Добавить элемент в хвост очереди; 

Удалить элемент из головы очереди; 

Проверка на отсутствие элементов (пустая очередь); 

Узнать значение информационной части первого элемента. 

Для работы с элементами очереди используется перемещение элементов из головы в хвост, с 

запоминанием адреса прежнего конца/начала очереди, чтобы не зациклиться, или использование 

дополнительной очереди. 

Дек (двухконечная очередь)  

(Информационно – логическая структура) 

(например, на основе двусвязного линейного списка + порядок доступа LIFO или 

FILO) 

DEQ = Double Ended Queue 

 
Данные: 

Адрес головы (конца дека, играющего роль головы в данный момент),  

Адрес хвоста (конца дека, играющего роль хвоста в данный момент). 

Допустимые операции: 

Создать дек из 1 элемента; 

Поменять роль голова-хвост (поменять направление двухконечной очереди); 

Добавить элемент в хвост дека; 

Удалить элемент из головы дека; 

Проверка на отсутствие элементов; 

Очистить дек (удалить все элементы) 

Узнать значение информационной головного элемента. 

Для работы с элементами дека используется перемещение элементов из головы в хвост, с 

запоминанием прежнего конца дека, чтобы не зациклиться; А также можно использовать 

дополнительный дек. 

 

Стек 

(Информационно – логическая структура) 

(например, на основе связного линейного списка + порядок доступа LIFO) 

 (LIFO - Last In First Out) 

 
Данные: 

Адрес вершины стека 

Допустимые операции: 

Создать стек и поместить в него 1 элемент; 

Добавить элемент в вершину стека; 

Удалить элемент из вершины стека; 

Проверка на отсутствие элементов (стек пуст); 

Очистить стек (удалить все элементы) 

Узнать значение информационной части элемента в вершине стека. 

Для работы с элементами стека используется дополнительный стек для временного хранения 

элементов просматриваемого стека. 

Бинарные деревья 

 

info 

left 

right 

info 

 

 

info 

Nil 

 

info 

Nil 

Nil 

info 

Данные: 

Адрес корня дерева. 

Допустимые операции: 

Создать дерево из 1 листа (элемент, без поддеревьев); 

Добавить лист/вершину/корень дерева (часто в соответствии с определенным на дереве 

порядком); 

Удалить лист/вершину/фрагмент дерева; 

Проверка на отсутствие элементов; 

Очистить дерево (удалить все элементы); 

Узнать значение информационной части текущего узла дерева. 

Для работы с элементами дерева используется обход дерева в глубину, в ширину, 

симметричный... Например, дерево с последнего в этом файле рисунка можно обойти следующими 

путями: 

 

10, 5, 1, 7, 6, 9, 10, 13 (начиная с корня в глубину слева направо) 

10, 5, 10, 1, 7, 13, 6, 9 (начиная с корня по ярусам) 

1, 6, 9, 7, 5, 13, 10, 10 (с самого левого листа, узлы после листьев) 

1, 5, 6, 7, 9, 10, 10, 13 (с самого левого листа проходя через узлы) 

и т.д. 

 

Дерево может быть построено, начиная с любого листа, в зависимости от нужд задачи. Например, 

при синтаксическом разборе выражения 5
96

1



 строится дерево, показанное ниже. 

 

 

На последнем рисунке приведен пример построения дерево с упорядоченными по возрастанию 

листьями. Любой лист, лежащий на дереве левее (и ниже) выбранного, меньше выбранного. А 

лист, лежащий выше и правее, больше или равен выбранному.  

Синтаксический разбор выражения 

Упорядоченные по возрастанию узлы 

Пример для лаб3 . Создать односвязный список, упорядоченный по убыванию абсолютной 

величины, из N вещественных чисел, лежащих в интервале [-9,9; 19,9]. Найти в нем максимальный 

элемент. Вывести созданный список или сообщение, что список пуст. Предусмотреть 1) 

возможность ввода данных в список из текстового файла; 2) генерации элементов в заданном 

количестве N; 3) добавления заданного элемента в список в указанном порядке в режиме диалога.  

 

Для работы со списком создадим отдельный модуль с процедурами: 

unit UnList; 

interface 

 

type 

  TInfo = real; 

 

  PElem = ^TElem; 

  TElem = record 

      info: TInfo; 

      next: PElem; 

  end; 

 

// создать список из одного элемента 

procedure CreateList(var ListN, ListK: PElem; r: TInfo); 

// добавить новый элемент перед первым 

procedure AddFirst(var ListN: PElem; ListK: PElem; r: TInfo); 

// добавить новый элемент в конец списка 

procedure AddLast(ListN: PElem; var ListK: PElem; r: TInfo); 

// добавить новый элемент в середину после ListC (не в конец) 

procedure AddMedium(ListN,ListC,ListK: PElem; r: TInfo); 

// очистить список 

procedure FreeList(var ListN, ListK: PElem); 

 

implementation 

 

// создать список из одного элемента 

procedure CreateList(var ListN, ListK: PElem; r: TInfo); 

begin 

  New(ListN); 

  ListK:=ListN; // конец и начало совпадают 

  ListN^.info:=r; 

  ListN^.next:=nil; 

end; 

 

// добавить новый элемент перед первым 

procedure AddFirst(var ListN: PElem; ListK: PElem; r: TInfo); 

var Elem: PElem; 

begin 

  new(Elem); 

  Elem^.info:=r; 

  Elem^.next:=ListN; 

  ListN:=Elem; // теперь он 1-ый 

end; 

// добавить новый элемент в конец 

procedure AddLast(ListN: PElem; var ListK: PElem; r: TInfo); 

begin 

  new(ListK^.next); 

  ListK^.next^.info:=r; 

  ListK^.next^.next:=nil; 

  ListK:= ListK^.next;  

end; 

 

 

// добавить новый элемент в середину после ListC (не в конец) 

procedure AddMedium(ListN,ListC,ListK: PElem; r: TInfo); 

var Elem: PElem; 

begin 

  new(Elem); 

  Elem^.info:=r; 

  Elem^.next:=ListC^.next; // после него то, что было после ListC 

  ListC^.next:=Elem; // а он сам после ListC 

end; 

 

// очистить список 

procedure FreeList(var ListN, ListK: PElem); 

var Elem: PElem; 

begin 

  Elem:=ListN; 

  while Elem<>nil do 

  begin 

    ListN:=ListN^.next; 

    Dispose(Elem); 

    Elem:=ListN; 

  end; 

  ListK:=nil; 

end; 

 

end. 

 

И модуль с глобальными переменными. В качестве примера использования необязательных 

разделов модуля (Initialization и Finalization), выполняемыми при подключении и отключении 

модуля, опишем в них инициализацию списка и освобождение выделенной памяти. 

unit UnVar; 

 

interface 

 

uses UnList;  // описание типа и все процедуры 

var NachaloSpiska, KonecSpiska: PElem; // начало и конец списка 

 

implementation 

  // пусто 

 

initialization 

  NachaloSpiska:=nil;  // инициализация переменных 

  KonecSpiska:=nil; 

finalization 

  FreeList(NachaloSpiska, KonecSpiska);  // освобождение памяти 

end. 

Головной модуль программы: 

program Lab4;  // один параметр - имя текстового файла 

 

{$APPTYPE CONSOLE} 

 

uses 

  SysUtils, 

  Windows, 

  UnList in 'UnList.pas',   // подключаем модули с типами и процедурами, а затем и с переменными 

  UnVar in 'UnVar.pas'; 

 

// добавить элемент со значением R в соответствии с заданным порядком в список 

procedure AddR(var ListN, ListK: PElem; r: TInfo); // адреса начала и конца списка и R 

var ListC: PElem;  // текущий элемент списка 

begin 

  if ListN = nil then CreateList(ListN, ListK, r)  // если список пуст 

  else 

    if abs(r) >= abs(ListN^.info) then AddFirst(ListN, ListK, r)  // добавить в начало 

    else 

       if abs(r) <= abs(ListK^.info) then AddLast(ListN, ListK, r)  // добавить в конец 

       else 

         begin // найти после какого 

           ListC:=ListN; // с начала списка 

           while abs(ListC^.next^.info) >= abs(r) do 

             ListC := ListC^.next; 

           AddMedium(ListN, ListC, ListK, r)  // добавить после текущего 

         end; 

end; 

 

//  "Добавить псевдослучайные N элементов" в текущий список 

procedure AddN; 

var n,i: integer; s:string; r: TInfo; 

begin 

  write('Сколько N=?'); readln(s); 

  if not TryStrToInt(s,n) then n:=1; 

  for i:=1 to n do 

  begin 

    r:= (-99 + Random(199+99+1))/10; 

    AddR(NachaloSpiska, KonecSpiska, r); 

  end; 

end; 

 

// Добавление всех данных из текстового файла в текущий список 

procedure AddFromTextFile; 

var 

  f: TextFile; 

  r: TInfo; 

begin 

  if ParamCount=1 then writeln('Введите элементы и Ctrl+Z') ; 

 

  AssignFile(f, ParamStr(1)); 

  try 

    Reset(f); 

    try 

      try 

          while not eof(f) do 

          begin 

            readln(f, r); 

            AddR(NachaloSpiska, KonecSpiska,r); 

          end; 

          writeln('Список создан'); 

      except 

       writeln('Некорректные данные в текстовом файле '+ ParamStr(1)); 

      end; 

    finally 

      CloseFile(f); 

    end; 

  except 

    writeln('Не удалось открыть текстовый файл '+ ParamStr(1)); 

  end; 

end; 

 

// Добавить один элемент  в текущий список 

procedure Add1; 

var 

  r: TInfo; 

begin 

 try 

  write('введите вещ.число R=?'); readln(r); 

  AddR(NachaloSpiska, KonecSpiska, r); 

 except 

    writeln('Некорректное число или ошибка добавления'); 

 end; 

end; 

 

// "Найти в списке" 

procedure SearchIt; 

var 

  ListC: PElem; 

begin 

         // сами - поиск и вывод результата 

         // для перебора элементов списка используйте цикл 

  ListC:=NachaloSpiska; 

  while ListC<>nil do   // или while (ListC<>nil) and not flag do 

  begin 

             // Здесь доступен текущий элемент и информация в нем ListC^.info 

    ListC:=ListC^.next; 

  end; 

 

end; 

 

//  "Что сейчас в списке?" - вывод на экран консоли 

procedure ViewList; 

var 

  ListC: PElem; 

begin 

  writeln('Список:'); 

  ListC:=NachaloSpiska; 

  if ListC=nil then writeln('Пустой'); 

  while ListC<>nil do 

  begin 

    writeln(ListC^.info:5:1); 

    ListC:=ListC^.next; 

  end; 

end; 

 

//--------------------главная программа------------------------- 

var 

  ch: char; 

begin 

  SetConsoleCP(1251);      // из модуля Windows 

  SetConsoleOutputCP(1251); 

 

{ Для того, чтобы каждый раз при запуске программы генерировалась 

  НОВАЯ последовательность псевдослучайных чисел. 

  Одна и та же последовательность удобна для отладки программы. } 

  Randomize; 

 

  repeat 

    writeln('--------------------------------------'); 

    writeln('Q - добавить в список из текстового файла; '); 

    writeln('W - сгенерировать и добавить в список; '); 

    writeln('E - добавить в список; '); 

    writeln('R - найти в списке;'); 

    writeln('T - удалить(очистить) список;'); 

    writeln('Y - просмотр;'); 

    writeln('U - конец.'); 

    write('Ваш выбор?'); readln(ch); 

    writeln('--------------------------------------'); 

 

    ch:= UpCase(ch); 

    case ch of 

//----------создать список из текстового файла------------------ 

    'Q': AddFromTextFile; 

//----------сгенерировать список ------------------------------- 

    'W': AddN; 

//----------добавить в список ---------------------------------- 

    'E': Add1; 

//-----------поиск---------------------------------------------- 

    'R': SearchIt; 

//-----------освобождение памяти-------------------------------- 

    'T': FreeList(NachaloSpiska, KonecSpiska); 

//-----------просмотр---------------------------------------------- 

    'Y': ViewList; 

//-----------выход---------------------------------------------- 

    'U': exit; 

//-------------------------------------------------------------- 

      else 

        begin 

          writeln('Нет такой команды'); 

          write('Press ENTER'); readln; 

        end; 

    end; 

 

  until ch='U'; 

end. 

 


